



DECUS

PROGRAM LIBRARY

DECUS NO.	8-362
TITLE	IOFMAG
AUTHOR	Giles Peterson
COMPANY	Submitted by: Ernest Hayden Speech Communications Research Laboratory Santa Barbara, California
DATE	December 1, 1970
SOURCE LANGUAGE	PAL III

Four subroutines are provided which access DECTapes on TC-01 drives. In contrast to the routines provided by DEC, these routines do not use the interrupt. If the user does not require the use of the interrupt while doing DECTape I/O, the use of these routines has several advantages:

- 1) They occupy less storage (106 [10] vs. 128 [10] words).
- 2) The user need not establish page \emptyset linkages to service the interrupt.
- 3) Other devices (such as the teletype) will not interrupt the DECTape I/O. When, for instance, transferring from paper tape to DECTape, one need not buffer those characters which would have been read during DECTape output.

The subroutine linkages are as follows:

```
JMS DTREA/read, ac must =  $\emptyset$ 
xxxx      /1st transfer location
u $\emptyset\emptyset\emptyset$   /u = unit #
-n        /n = # of blocks to be transferred
b         /b = 1st block to be transferred
xxxx      /instruction executed when read completed
```

```
JMS DTWRI/write, ac must =  $\emptyset$ 
xxxx      /1st transfer location
u $\emptyset\emptyset\emptyset$   /u = unit #
-n        /n = # of blocks to be transferred
b         /b = 1st block to be transferred
xxxx      /instruction executed when write completed
```

```
JMS DTSAS/search and stop, ac must = block # wanted
u $\emptyset\emptyset\emptyset$   /u = unit #
xxxx      /instruction executed when search and stop completed
```

```
JMS DTSNS/search and don't stop, ac must =  $\emptyset$ 
    DTBLOC must be put = block # wanted
    DTUNIT must be put = u $\emptyset\emptyset\emptyset$ 
    note: the other subs will set DTBLOC & DTUNIT.
```

If a tape error occurs, the sub will go to the address specified in LDERR which is initially 76 $\emptyset\emptyset$.

The block length is specified in DT76 $\emptyset\emptyset$, which is initially 76 $\emptyset\emptyset$ (-2 $\emptyset\emptyset$).

The DTSNS routine differs from DTSAS only in a) linkages and b) the tape is in motion when the subroutine is completed.

7525	7006	RTL
7526	7200	CLA
7527	1367	TAD DTBLO
7530	7041	CIA
7531	1370	TAD DTBLOC
7532	7450	SNA
7533	5350	JMP DTSNS5
7534	7041	CIA /IF FORE & < OR REV & > OR REV & WANTED -1, SKIP
7535	7420	SNL
7536	7001	IAC
7537	7620	SNL CLA
7540	1364	DTSNS2, TAD DT0400 /SET TO TURN AROUND
7541	6764	DTXA /IF SET, TURN; & CLEAR FLAGS
7542	5312	JMP DTSNS1 /GO TO DTSNS1
7543	1363	DTSNS3, TAD DT0200 /SET TO TURN AROUND IN END ZONE
7544	5340	JMP DTSNS2 /GO TO DTSNS2
7545	1373	DTSNS4, TAD DTUNIT /STOP UNIT
7546	6766	DTLA
7547	5776	JMP I LDERR /GO TO DERR
7550	6764	DTSNS5, DTXA /CLEAR FLAGS
7551	7620	SNL CLA /IF REVERSE, GO TO DTSNS1
7552	5312	JMP DTSNS1
7553	5704	JMP I DTSNS /END OF SUB
7554	0000	DTWRI, 0 /WRITE SUB
7555	4254	JMS DTGET /GET ARGUMENTS
7556	4304	JMS DTSNS /SEARCH & DON'T STOP
7557	1362	TAD DT0050 /SET TO WRITE
7560	5234	JMP DTGO /GO TO DTGO
7561	0030	DT0030, 30
7562	0050	DT0050, 50
7563	0200	DT0200, 200
7564	0400	DT0400, 400
7565	0610	DT0610, 610
7566	7600	DT7600, 7600
7567	0000	DTBLO, 0
7570	0000	DTBLOC, 0
7571	0000	DTLOC, 0
7572	0000	DTMNBL, 0
7573	0000	DTUNIT, 0
7574	7567	LDTBLO, DTBLO
7575	7754	LDTC, 7754
7576	7600	LDERR, 7600
7577	7755	LDTL, 7755

*7426 /IOFMAG 13-50

7426	0000	DTREA, 0 /READ SUB
7427	1226	TAD DTREA /SET SUB END
7430	3354	DCA DTWRI
7431	4254	JMS DTGET /GET ARGUMENTS
7432	4304	JMS DTSNS /SEARCH & DON'T STOP
7433	1361	TAD DT0030 /SET TO READ
7434	6764	DTGO, DTXA /SET FOR READ OR WRITE
7435	1371	TAD DTLOC
7436	3777	DCA I LDTL
7437	6764	DTGO1, DTXA /START 128 TRANSFERS & CLEAR FLAGS
7440	1366	TAD DT7600
7441	3775	DCA I LDTC
7442	6771	DTSF /WAIT FOR FLAG


```

7443 5242    JMP .-1
7444 6772    DTRB /IF ERROR, GO TO DTSNS4
7445 7710    SPA CLA
7446 5345    JMP DTSNS4
7447 2372    ISZ DTMNBL /IF MORE BLOCKS, GO TO DTG01
7450 5237    JMP DTG01
7451 1373    TAD DTUNIT /STOP UNIT
7452 6766    DTLA
7453 5754    JMP I DTWRI /END OF SUB
7454 0000    DTGET, 0 /GET ARGUMENTS SUB
7455 7040    CMA /GET TRANSFER LOC
7456 4267    JMS DTGET1
7457 3371    DCA DTLOC
7460 4267    JMS DTGET1 /GET UNIT #
7461 3373    DCA DTUNIT
7462 4267    JMS DTGET1 /GET - # OF BLOCKS
7463 3372    DCA DTMNBL
7464 4267    JMS DTGET1 /GET BLOCK #
7465 3370    DCA DTBLOC
7466 5654    JMP I DTGET
7467 0000    DTGET1, 0 /GET NEXT ARGUMENT SUB
7470 1754    TAD I DTWRI
7471 2354    ISZ DTWRI
7472 5667    JMP I DTGET1
7473 0000    DTSAS, 0 /SEARCH & STOP SUB
7474 3370    DCA DTBLOC /PUT DTBLOC = BLOCK #
7475 1673    TAD I DTSAS /GET UNIT #
7476 2273    ISZ DTSAS
7477 3373    DCA DTUNIT
7500 4304    JMS DTSNS /SEARCH & DON'T STOP
7501 1373    TAD DTUNIT /STOP TAPE
7502 6766    DTLA
7503 5673    JMP I DTSAS /END OF SUB
7504 0000    DTSNS, 0 /SEARCH & DON'T STOP SUB
7505 1374    TAD LDTBLO /START NORMAL REV SEARCH
7506 3777    DCA I LDTL
7507 1373    TAD DTUNIT
7510 1365    TAD DT0610
7511 6766    DTLA
7512 6771    DTSNS1, DTSF /WAIT FOR FLAG
7513 5312    JMP .-1
7514 6772    DTRB /IF IN END ZONE, GO TO DTSNS3
7515 7006    RTL
7516 7710    SPA CLA
7517 5343    JMP DTSNS3
7520 6772    DTRB /IF ERROR, GO TO DTSNS4
7521 7710    SPA CLA
7522 5345    JMP DTSNS4
7523 6761    DTRA /IF BLOCK FOUND, GO TO DTSNS5
7524 7006    RTL

```

